Rules for integrands involving zeta functions

1. $\int \text{Zeta}[s,\ a+b\,x]\,dx$

1: $\int \text{Zeta}[2,\ a+b\,x]\,dx$

Derivation: Algebraic simplification

Basis: $\zeta(2,z) = \psi^{(1)}(z)$

Rule:

$$\int \text{Zeta}[2,\ a+b\,x]\,dx \longrightarrow \int \text{PolyGamma}[1,\ a+b\,x]\,dx$$

Program code:

```
Int[Zeta[2,a_.+b_.*x_],x_Symbol] :=
  Int[PolyGamma[1,a+b*x],x] /;
FreeQ[{a,b},x]
```

2: $\int \text{Zeta}[s, a + b x] \, dx$ when $s \neq 1 \land s \neq 2$

Derivation: Primitive rule

Basis: $\frac{\partial \zeta(s,z)}{\partial z} = -s \, \zeta(s+1, z)$

Rule: If $s \neq 1 \land s \neq 2$, then

$$\int \text{Zeta}[s, a + b x] \, dx \longrightarrow -\frac{\text{Zeta}[s-1, a+b x]}{b \, (s-1)}$$

Program code:

```
Int[Zeta[s_,a_.+b_.*x_],x_Symbol] :=
  -Zeta[s-1,a+b*x]/(b*(s-1)) /;
FreeQ[{a,b,s},x] && NeQ[s,1] && NeQ[s,2]
```

2. $\int (c + d\,x)^m\,\text{Zeta}[s,\,a + b\,x]\,dx$

1: $\int (c + d\,x)^m\,\text{Zeta}[2,\,a + b\,x]\,dx$ when $m \in \mathbb{Q}$

Derivation: Algebraic simplification

Basis: $\zeta(2, z) = \psi^{(1)}(z)$

Rule: If $m \in \mathbb{Q}$, then

$$\int (c + d\,x)^m\,\text{Zeta}[2,\,a + b\,x]\,dx \longrightarrow \int (c + d\,x)^m\,\text{PolyGamma}[1,\,a + b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Zeta[2,a_.+b_.*x_],x_Symbol] :=
  Int[(c+d*x)^m*PolyGamma[1,a+b*x],x] /;
FreeQ[{a,b,c,d},x] && RationalQ[m]
```

2. $\int (c + d x)^m \text{Zeta}[s, a + b x] \, dx$ when $s \neq 1 \land s \neq 2$

1: $\int (c + d x)^m \text{Zeta}[s, a + b x] \, dx$ when $s \neq 1 \land s \neq 2 \land m > 0$

Derivation: Integration by parts

Rule: If $s \neq 1 \land s \neq 2 \land m > 0$, then

$$\int (c + d x)^m \text{Zeta}[s, a + b x] \, dx \longrightarrow -\frac{(c + d x)^m \text{Zeta}[s - 1, a + b x]}{b (s - 1)} + \frac{d m}{b (s - 1)} \int (c + d x)^{m-1} \text{Zeta}[s - 1, a + b x] \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Zeta[s_,a_.+b_.*x_],x_Symbol] :=
  -(c+d*x)^m*Zeta[s-1,a+b*x]/(b*(s-1)) +
  d*m/(b*(s-1))*Int[(c+d*x)^(m-1)*Zeta[s-1,a+b*x],x] /;
FreeQ[{a,b,c,d,s},x] && NeQ[s,1] && NeQ[s,2] && GtQ[m,0]
```

2: $\int (c + d x)^m \text{Zeta}[s, a + b x] \, dx$ when $s \neq 1 \land s \neq 2 \land m < -1$

Derivation: Inverted integration by parts

Rule: If $s \neq 1 \land s \neq 2 \land m < -1$, then

$$\int (c + d x)^m \text{Zeta}[s, a + b x] \, dx \longrightarrow \frac{(c + d x)^{m+1} \text{Zeta}[s, a + b x]}{d (m + 1)} + \frac{b s}{d (m + 1)} \int (c + d x)^{m+1} \text{Zeta}[s + 1, a + b x] \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Zeta[s_,a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*Zeta[s,a+b*x]/(d*(m+1)) +
  b*s/(d*(m+1))*Int[(c+d*x)^(m+1)*Zeta[s+1,a+b*x],x] /;
FreeQ[{a,b,c,d,s},x] && NeQ[s,1] && NeQ[s,2] && LtQ[m,-1]
```